
Subject: Re: Merging Birth's Recode and Household Member Recode

Posted by [Bridgette-DHS](#) on Wed, 10 Feb 2016 15:38:34 GMT

[View Forum Message](#) <> [Reply to Message](#)

Following is a response from Senior DHS Stata Specialist, Tom Pullum:

We apologize for the long delay in responding to this question from November 30. The strategy you propose, for merging the PR and BR files when b16 is not in the BR file, would work most of the time, but not always. The main problem is that age in the PR file (hv105) and age in the BR file (b8) do not always agree. Below I show a way to assess this problem when b16 is present, using the Ghana 2014 survey. You could also approach it by doing the merge as you propose and then checking b16. But if you really do not have b16 and you use your approach, you will get some false matches and some failed matches.

- * Possible strategy to match children age 0-17 in the PR and BR files when b16 is not available.
- * The proposed criteria use the age and sex of the child.
- * Test on a file for which b16 IS available, and look for correspondences in the proposed criteria.

- * Children will only appear in the PR file if their mother is in the PR file.
- * Assume we have hv112, but actually it is not included in all surveys.

- * sex: hv104 and b4
- * age in years: hv105 and b8 (NOTE: do not always agree, even for genuine matches)
- * line number in the PR file: hvidx and b16

set more off

use e:\DHS\DHS_data\BR_files\GHBR70FL.dta, clear

drop if b5==0

keep v000 v001 v002 v003 b4 b8 b16

rename v000 hv000

rename v001 hv001

rename v002 hv002

rename b16 hvidx

sort hv000 hv001 hv002 hvidx

save e:\DHS\DHS_data\scratch\BRtemp.dta, replace

use e:\DHS\DHS_data\PR_files\GHPR70FL.dta, clear

drop if hv112==0 | hv112==.

keep hv000 hv001 hv002 hvidx hv104 hv105

sort hv000 hv001 hv002 hvidx

save e:\DHS\DHS_data\scratch\PRtemp.dta, replace

merge hv000 hv001 hv002 hvidx using e:\DHS\DHS_data\scratch\BRtemp.dta

tab _merge

gen sex_agree=0

replace sex_agree=1 if hv104==b4

tab sex_agree _merge,m

* assume that hv104 and b4 will always agree; this is valid because it is equivalent to _merge=3.
keep if sex_agree==1

```
gen age_agree=0
replace age_agree=1 if hv105==b8
tab age_agree _merge,m
```

```
tab hv105 age_agree, row
```

* The main problem is that hv105 and b8 do not agree for about 8% of children 0-17 (and 6% of children 0-4)

If you apply this same check in other surveys that include b16, such as the Liberia 2013 DHS survey, you find a higher level of disagreement between hv105 and b8. In most of the surveys that do not include b16, and therefore are earlier surveys, merges using the sex and age of the child will be even worse. We strongly recommend that you do not try to merge the PR file with the BR or KR file when b16 is not in the data.
